Curvature and the geodesic equations

The code in this notebook is based on James Hartle's Mathematica programs written as companions to his book "Gravity: An Introduction to Einstein's General Relativity", available from the Book's site: http://web.physics.ucsb.edu/~gravitybook/mathematica.html
The original program was written by *Leonard Parker, University of Wisconsin, Milwaukee* (see the Acknowledgement section in the end of this notebook). The code in this notebook is a modification of the original program, adapted for the needs of the course "General Relativity and Cosmology" offered at the National Technical University of Athens, by Konstantinos Anagnostopoulos (http://physics.ntua.gr/konstant).The site of the course for the Spring 2023 can be found at *http://physics.ntua.gr/konstant/GR* .

This is the *Mathematica* notebook *Curvature and the Einstein Equation* available from the book website. From a given metric $g_{\alpha\beta}$, it computes the components of the following: the inverse metric, $g^{\lambda\sigma}$, the Christoffel symbols or affine connection,

$\Gamma^{\lambda}{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_{\mu} g_{\sigma\nu} + \partial_{\nu} g_{\sigma\mu} - \partial_{\sigma} g_{\mu\nu})$,

(∂_{α} stands for the partial derivative $\partial / \partial x^{\alpha}$), the Riemann tensor,

$R^{\lambda}{}_{\mu\nu\sigma} = \partial_{\nu} \Gamma^{\lambda}{}_{\mu\sigma} - \partial_{\sigma} \Gamma^{\lambda}{}_{\mu\nu} + \Gamma^{\lambda}{}_{\eta\nu} \Gamma^{\eta}{}_{\mu\sigma} - \Gamma^{\lambda}{}_{\eta\sigma} \Gamma^{\eta}{}_{\mu\nu}$,

the Ricci tensor

$R_{\mu\nu} = R^{\lambda}{}_{\mu\lambda\nu}$,

the scalar curvature,

$R = g^{\mu\nu} R_{\mu\nu}$,

and the Einstein tensor,

$G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R$.

We also compute:

$R_{\mu\nu\rho\sigma}$, $R^{\mu\nu\rho\sigma}$, $R^2 = R_{\mu\nu\rho\sigma} R^{\mu\nu\rho\sigma}$, the Weyl Tensor

$C_{\rho\sigma\mu\nu} = R_{\rho\sigma\mu\nu} - \frac{2}{n-2} \{g_{\rho[\mu} R_{\nu]\sigma} - g_{\sigma[\mu} R_{\nu]\rho}\} + \frac{2}{(n-1)(n-2)} g_{\rho[\mu} g_{\nu]\sigma} R$,

and the geodesic equations.

You must input the covariant components of the metric tensor $g_{\mu\nu}$ by editing the relevant input line in this *Mathematica* notebook. You may also wish to change the names of the coordinates. Only the nonzero components of the above quantities are displayed as the output. All the components computed are in the *coordinate basis* in which the metric was specified.

Initialization:

First clear any values that may already have been assigned to the names of the various objects to be calculated. The names of the coordinates that you will use are also cleared. The name of the coordinates must be defined (the list **coord**), and then the dimension of the manifold **n** is set.

Note:
It is important not to use the symbols, **i**, **j**, **k**, **l, s, R, G, C, τ, Γ,** or **n** as constants or coordinates in the metric that you specify above. The reason is that the first five of those symbols are used as summation or table indices in the calculations done below, and **n** is the dimension of the space. The rest are also used for displaying results in the program.

```
Clear[coord, metric, inversemetric, affine, riemann, lriemann, uriemann,
  ricci, scalar, einstein, weyl, geodesic, R, G, τ, i, j, k, l, s];
Clear[r, θ, ϕ, t, χ, a, m];

(*----------------------------------------------------------------------------*)
(* This is what you need to set + the metric below: *)
coord = {t, χ, θ, ϕ};
n     = Length[coord];
(*----------------------------------------------------------------------------*)
Print["The Manifold has dimension   n= ",
 n, "\nCoordinate system:          ", coord]
```

```
The Manifold has dimension   n= 4
Coordinate system:          {t, χ, θ, ϕ}
```

Defining the metric: The Friedmann metric

Input the metric as a list of lists, i.e., as a matrix. You can input the components of any metric here, but you must specify them as explicit functions of the coordinates.

Compute for the metric:
$ds^2 = -dt^2 + a(t)^2\left[\,d\chi^2 + \sin^2(\chi)\left(d\theta^2 + \sin^2(\theta)\, d\phi^2\right)\right]$

In[ ]:=

```
(*------------------------------------------------------------------------------*)
metric = {
    {-1, 0    , 0              ,                    0},
    {0, a[t]^2, 0              ,                    0},
    {0, 0     , a[t]^2 Sin[χ]^2,                    0},
    {0, 0     , 0              , a[t]^2 Sin[χ]^2 Sin[θ]^2}};
(*------------------------------------------------------------------------------*)
metric // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & a[t]^2 & 0 & 0 \\ 0 & 0 & a[t]^2\,\mathrm{Sin}[\chi]^2 & 0 \\ 0 & 0 & 0 & a[t]^2\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2 \end{pmatrix}$$

Calculating the inverse metric:

The inverse metric is obtained through matrix inversion.

In[ ]:=

```
inversemetric = FullSimplify[Inverse[metric]];
inversemetric // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & \frac{1}{a[t]^2} & 0 & 0 \\ 0 & 0 & \frac{\mathrm{Csc}[\chi]^2}{a[t]^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2\,\mathrm{Csc}[\chi]^2}{a[t]^2} \end{pmatrix}$$

Let’s see how we can compute the derivatives:

In[ ]:=

```
metric[[2, 2]]
```

Out[ ]= $a[t]^2$

In[ ]:=

```
coord[[1]]
```

Out[ ]= t

In[ ]:=

```
D[metric[[2, 2]], t]
```

Out[ ]= 2 a[t] a′[t]

Use the list of coordinates to specify the variable w.r.t. which we differentiate:

In[ ]:=
```
D[metric[[2, 2]], coord[[1]]]
```
Out[ ]=
```
2 a[t] a′[t]
```

In[ ]:=
```
Table[D[metric[[i, i]], coord[[1]]], {i, n}]
```
Out[ ]=
```
{0, 2 a[t] a′[t], 2 a[t] Sin[χ]^2 a′[t], 2 a[t] Sin[θ]^2 Sin[χ]^2 a′[t]}
```

In[ ]:=
```
Table[D[metric[[i, i]], coord[[j]]], {i, n}, {j, n}]
```
Out[ ]=
```
{{0, 0, 0, 0}, {2 a[t] a′[t], 0, 0, 0}, {2 a[t] Sin[χ]^2 a′[t], 2 a[t]^2 Cos[χ] Sin[χ], 0, 0},
 {2 a[t] Sin[θ]^2 Sin[χ]^2 a′[t], 2 a[t]^2 Cos[χ] Sin[θ]^2 Sin[χ], 2 a[t]^2 Cos[θ] Sin[θ] Sin[χ]^2, 0}}
```

In[ ]:=
```
∂_t metric[[2, 2]]
```
Out[ ]=
```
2 a[t] a′[t]
```

In[ ]:=
```
∂_coord[[1]] metric[[2, 2]]
```
Out[ ]=
```
2 a[t] a′[t]
```

Calculating the Christoffel symbols:

The calculation of the components of the Christoffel symbols is done by transcribing the definition given earlier into the notation of *Mathematica* and using the *Mathematica* functions **D** for taking partial derivatives, **Sum** for summing over repeated indices, **Table** for forming a list of components, and **Simplify** for simplifying the result.

The delayed definition performs a calculation on demand and caches the result:
E.g.: pi2:=pi2=N[Pi^2,50];
fib[1]=fib[2]=1;
fib[n_]:=fib[n]=fib[n-1]+fib[n-2]
fib[6] → 8
Definition[fib]

```
fib[1] = 1
fib[2] = 1
fib[3] = 2
fib[4] = 3
fib[5] = 5
fib[6] = 8
fib[n_] := fib[n] = fib[n - 1] + fib[n - 2]
```

In[ ]:=
```
affine := affine = FullSimplify[Table[
    (1 / 2) * Sum[
      (*              g^is    (  ∂_k g_sj                 +
         ∂_j g_sk                     -     ∂_s g_jk)              *)
      (inversemetric[[i, s]]) * (D[metric[[s, j]], coord[[k]]] +
         D[metric[[s, k]], coord[[j]]] - D[metric[[j, k]], coord[[s]]]),
      {s, 1, n}],
    {i, 1, n}, {j, 1, n}, {k, 1, n}]];
```

Now let's see how to display the nontrivial results. First let's look what affine consists of:

In[ ]:=
```
affine
```

Out[ ]=

$$\left\{\left\{\{0,0,0,0\},\{0,a[t]\,a'[t],0,0\},\{0,0,a[t]\,\mathrm{Sin}[\chi]^2\,a'[t],0\},\{0,0,0,a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a'[t]\}\right\},\left\{\left\{0,\frac{a'[t]}{a[t]},0,0\right\},\left\{\frac{a'[t]}{a[t]},0,0,0\right\},\{0,0,-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi],0\},\{0,0,0,-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]\}\right\},\left\{\left\{0,0,\frac{a'[t]}{a[t]},0\right\},\{0,0,\mathrm{Cot}[\chi],0\},\left\{\frac{a'[t]}{a[t]},\mathrm{Cot}[\chi],0,0\right\},\{0,0,0,-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\}\right\},\left\{\left\{0,0,0,\frac{a'[t]}{a[t]}\right\},\{0,0,0,\mathrm{Cot}[\chi]\},\{0,0,0,\mathrm{Cot}[\theta]\},\left\{\frac{a'[t]}{a[t]},\mathrm{Cot}[\chi],\mathrm{Cot}[\theta],0\right\}\right\}\right\}$$

We can look at $\Gamma^{\mu}{}_{\nu\rho}$ as matrices $(\Gamma^{\mu})_{\nu\rho}$:

In[ ]:=
```
Table[affine[[i]] // MatrixForm, {i, 1, n}]
```

Out[ ]=

$$\left\{\begin{pmatrix}0&0&0&0\\0&a[t]\,a'[t]&0&0\\0&0&a[t]\,\mathrm{Sin}[\chi]^2\,a'[t]&0\\0&0&0&a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a'[t]\end{pmatrix},\begin{pmatrix}0&\frac{a'[t]}{a[t]}&0&0\\\frac{a'[t]}{a[t]}&0&0&0\\0&0&-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi]&0\\0&0&0&-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]\end{pmatrix},\begin{pmatrix}0&0&\frac{a'[t]}{a[t]}&0\\0&0&\mathrm{Cot}[\chi]&0\\\frac{a'[t]}{a[t]}&\mathrm{Cot}[\chi]&0&0\\0&0&0&-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\end{pmatrix},\begin{pmatrix}0&0&0&\frac{a'[t]}{a[t]}\\0&0&0&\mathrm{Cot}[\chi]\\0&0&0&\mathrm{Cot}[\theta]\\\frac{a'[t]}{a[t]}&\mathrm{Cot}[\chi]&\mathrm{Cot}[\theta]&0\end{pmatrix}\right\}$$

Many zeroes... We also need to associate the nonzero values with the indices μ, ν, ρ. Let's make pairs of values and $\Gamma^{\mu}{}_{\nu\rho}$ symbols:

We use the functions Superscript[Γ, *i*] → Γ^{i}, and Subscript[*A*, *j*, *k*] → $A_{j,k}$. The function UnsameQ[<expression>,0] evaluates to True, when <expression> is not the same as 0. If[<condition>,<value>] returns <value> when <condition> is True and Null when it is False:

In[ ]:=
```
listaffine := Table[
   If[
    UnsameQ[affine⟦i, j, k⟧, 0],
    {Subscript[Superscript[Γ, i], j, k], affine⟦i, j, k⟧}
    ],
   {i, 1, n}, {j, 1, n}, {k, 1, j}];
```

Now let's see what listaffine consists of:

In[ ]:=
```
listaffine
```

Out[ ]=
$$\left\{\{\{\text{Null}\}, \{\text{Null}, \{\Gamma^{1}{}_{2,2}, a[t]\,a'[t]\}\}, \{\text{Null}, \text{Null}, \{\Gamma^{1}{}_{3,3}, a[t]\,\text{Sin}[\chi]^2\,a'[t]\}\}, \{\text{Null}, \text{Null}, \text{Null}, \{\Gamma^{1}{}_{4,4}, a[t]\,\text{Sin}[\theta]^2\,\text{Sin}[\chi]^2\,a'[t]\}\}\}, \left\{\{\text{Null}\}, \left\{\left\{\Gamma^{2}{}_{2,1}, \frac{a'[t]}{a[t]}\right\}, \text{Null}\right\}, \{\text{Null}, \text{Null}, \{\Gamma^{2}{}_{3,3}, -\text{Cos}[\chi]\,\text{Sin}[\chi]\}\}, \{\text{Null}, \text{Null}, \text{Null}, \{\Gamma^{2}{}_{4,4}, -\text{Cos}[\chi]\,\text{Sin}[\theta]^2\,\text{Sin}[\chi]\}\}\right\}, \left\{\{\text{Null}\}, \{\text{Null}, \text{Null}\}, \left\{\left\{\Gamma^{3}{}_{3,1}, \frac{a'[t]}{a[t]}\right\}, \{\Gamma^{3}{}_{3,2}, \text{Cot}[\chi]\}, \text{Null}\right\}, \{\text{Null}, \text{Null}, \text{Null}, \{\Gamma^{3}{}_{4,4}, -\text{Cos}[\theta]\,\text{Sin}[\theta]\}\}\right\}, \left\{\{\text{Null}\}, \{\text{Null}, \text{Null}\}, \{\text{Null}, \text{Null}, \text{Null}\}, \left\{\left\{\Gamma^{4}{}_{4,1}, \frac{a'[t]}{a[t]}\right\}, \{\Gamma^{4}{}_{4,2}, \text{Cot}[\chi]\}, \{\Gamma^{4}{}_{4,3}, \text{Cot}[\theta]\}, \text{Null}\right\}\right\}\right\}$$

Now we want to get rid of the Nulls: First Flatten the list:

In[ ]:=
```
Flatten[listaffine]
```

Out[ ]=
$$\left\{\text{Null}, \text{Null}, \Gamma^{1}{}_{2,2}, a[t]\,a'[t], \text{Null}, \text{Null}, \Gamma^{1}{}_{3,3}, a[t]\,\text{Sin}[\chi]^2\,a'[t], \text{Null}, \text{Null}, \text{Null}, \Gamma^{1}{}_{4,4}, a[t]\,\text{Sin}[\theta]^2\,\text{Sin}[\chi]^2\,a'[t], \text{Null}, \Gamma^{2}{}_{2,1}, \frac{a'[t]}{a[t]}, \text{Null}, \text{Null}, \text{Null}, \Gamma^{2}{}_{3,3}, -\text{Cos}[\chi]\,\text{Sin}[\chi], \text{Null}, \text{Null}, \text{Null}, \Gamma^{2}{}_{4,4}, -\text{Cos}[\chi]\,\text{Sin}[\theta]^2\,\text{Sin}[\chi], \text{Null}, \text{Null}, \text{Null}, \Gamma^{3}{}_{3,1}, \frac{a'[t]}{a[t]}, \Gamma^{3}{}_{3,2}, \text{Cot}[\chi], \text{Null}, \text{Null}, \text{Null}, \text{Null}, \Gamma^{3}{}_{4,4}, -\text{Cos}[\theta]\,\text{Sin}[\theta], \text{Null}, \text{Null}, \text{Null}, \text{Null}, \text{Null}, \text{Null}, \Gamma^{4}{}_{4,1}, \frac{a'[t]}{a[t]}, \Gamma^{4}{}_{4,2}, \text{Cot}[\chi], \Gamma^{4}{}_{4,3}, \text{Cot}[\theta], \text{Null}\right\}$$

Then get rid of the Nulls, using the DelteCases function:

In[ ]:=
```
DeleteCases[Flatten[listaffine], Null]
```

Out[ ]=

$$\left\{\Gamma^1{}_{2,2},\ a[t]\,a'[t],\ \Gamma^1{}_{3,3},\ a[t]\,\mathrm{Sin}[\chi]^2\,a'[t],\ \Gamma^1{}_{4,4},\ a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a'[t],\ \Gamma^2{}_{2,1},\ \frac{a'[t]}{a[t]},\ \Gamma^2{}_{3,3},\ -\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi],\ \Gamma^2{}_{4,4},\ -\mathrm{Cos}[\chi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi],\ \Gamma^3{}_{3,1},\ \frac{a'[t]}{a[t]},\ \Gamma^3{}_{3,2},\ \mathrm{Cot}[\chi],\ \Gamma^3{}_{4,4},\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta],\ \Gamma^4{}_{4,1},\ \frac{a'[t]}{a[t]},\ \Gamma^4{}_{4,2},\ \mathrm{Cot}[\chi],\ \Gamma^4{}_{4,3},\ \mathrm{Cot}[\theta]\right\}$$

Then, reconstruct pairs of {Symbol,Values} using the Partition function:

In[ ]:=
```
Partition[DeleteCases[Flatten[listaffine], Null], 2]
```

Out[ ]=

$$\left\{\{\Gamma^1{}_{2,2},\ a[t]\,a'[t]\},\ \{\Gamma^1{}_{3,3},\ a[t]\,\mathrm{Sin}[\chi]^2\,a'[t]\},\ \{\Gamma^1{}_{4,4},\ a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a'[t]\},\ \left\{\Gamma^2{}_{2,1},\ \frac{a'[t]}{a[t]}\right\},\ \{\Gamma^2{}_{3,3},\ -\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi]\},\ \{\Gamma^2{}_{4,4},\ -\mathrm{Cos}[\chi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]\},\ \left\{\Gamma^3{}_{3,1},\ \frac{a'[t]}{a[t]}\right\},\ \{\Gamma^3{}_{3,2},\ \mathrm{Cot}[\chi]\},\ \{\Gamma^3{}_{4,4},\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\},\ \left\{\Gamma^4{}_{4,1},\ \frac{a'[t]}{a[t]}\right\},\ \{\Gamma^4{}_{4,2},\ \mathrm{Cot}[\chi]\},\ \{\Gamma^4{}_{4,3},\ \mathrm{Cot}[\theta]\}\right\}$$

Yeah, we are ready, now let’s make a pretty display:

In[ ]:=
```
Partition[DeleteCases[Flatten[listaffine], Null], 2] // TableForm
```

Out[ ]//TableForm=

$\Gamma^1{}_{2,2}$	$a[t]\,a'[t]$
$\Gamma^1{}_{3,3}$	$a[t]\,\mathrm{Sin}[\chi]^2\,a'[t]$
$\Gamma^1{}_{4,4}$	$a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a'[t]$
$\Gamma^2{}_{2,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^2{}_{3,3}$	$-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi]$
$\Gamma^2{}_{4,4}$	$-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]$
$\Gamma^3{}_{3,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^3{}_{3,2}$	$\mathrm{Cot}[\chi]$
$\Gamma^3{}_{4,4}$	$-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]$
$\Gamma^4{}_{4,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^4{}_{4,2}$	$\mathrm{Cot}[\chi]$
$\Gamma^4{}_{4,3}$	$\mathrm{Cot}[\theta]$

Or use more spacing for more comfortable reading: TableSpacing → {yspaces, xspaces} adds spaces in rows+columns

In[ ]:=
```
TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2], TableSpacing → {4, 8}]
```

Out[ ]//TableForm=

$\Gamma^{1}{}_{2,2}$	a[t] a′[t]
$\Gamma^{1}{}_{3,3}$	a[t] Sin[χ]^2 a′[t]
$\Gamma^{1}{}_{4,4}$	a[t] Sin[θ]^2 Sin[χ]^2 a′[t]
$\Gamma^{2}{}_{2,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^{2}{}_{3,3}$	−Cos[χ] Sin[χ]
$\Gamma^{2}{}_{4,4}$	−Cos[χ] Sin[θ]^2 Sin[χ]
$\Gamma^{3}{}_{3,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^{3}{}_{3,2}$	Cot[χ]
$\Gamma^{3}{}_{4,4}$	−Cos[θ] Sin[θ]
$\Gamma^{4}{}_{4,1}$	$\frac{a'[t]}{a[t]}$
$\Gamma^{4}{}_{4,2}$	Cot[χ]
$\Gamma^{4}{}_{4,3}$	Cot[θ]

Calculating and displaying the Riemann tensor:

The components of the Riemann tensor, $R^{\lambda}{}_{\mu\nu\sigma}$, are calculated using the definition given above.

$R^{i}{}_{jkl} = \partial_k \Gamma^{i}{}_{lj} - \partial_l \Gamma^{i}{}_{kj} + \Gamma^{i}{}_{km} \Gamma^{m}{}_{lj} - \Gamma^{i}{}_{lm} \Gamma^{m}{}_{kj} = \Gamma^{i}{}_{lj;k} - \Gamma^{i}{}_{kj;l} + \Gamma^{i}{}_{ks} \Gamma^{s}{}_{lj} - \Gamma^{i}{}_{ls} \Gamma^{s}{}_{kj}$

In[ ]:=

```
riemann := riemann = FullSimplify[Table[
    (* Riᵢjkl=    ∂kΓilj                     -          ∂lΓikj*)
    D[   affine[[i, l, j]],  coord[[k]]    ] - D[affine[[i, k, j]], coord[[l]]    ] +
     (*          Γiks               Γslj     -          Γils              Γskj      *)
     Sum[affine[[i, k, s]] affine[[s, l, j]] -  affine[[i, l, s]] affine[[s, k, j]] ,
      {s, 1, n}],
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listriemann := Table[
   If[
    UnsameQ[riemann[[i, j, k, l]], 0],
    {Subscript[Superscript[R, i], j, k, l], riemann[[i, j, k, l]]}
   ],
   {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}];
TableForm[Partition[DeleteCases[Flatten[listriemann], Null], 2],
 TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

$R^1{}_{2,2,1}$	$-a[t]\, a''[t]$
$R^1{}_{3,3,1}$	$-a[t]\, \mathrm{Sin}[\chi]^2\, a''[t]$
$R^1{}_{4,4,1}$	$-a[t]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, a''[t]$
$R^2{}_{1,2,1}$	$-\frac{a''[t]}{a[t]}$
$R^2{}_{3,3,2}$	$-\mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$
$R^2{}_{4,4,2}$	$-\mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$
$R^3{}_{1,3,1}$	$-\frac{a''[t]}{a[t]}$
$R^3{}_{2,3,2}$	$1 + a'[t]^2$
$R^3{}_{4,4,3}$	$-\mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$
$R^4{}_{1,4,1}$	$-\frac{a''[t]}{a[t]}$
$R^4{}_{2,4,2}$	$1 + a'[t]^2$
$R^4{}_{3,4,3}$	$\mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$

In[ ]:=
```
Table[riemann[[i, j]] // MatrixForm, {i, n}, {j, n}]
```

Out[ ]=

$$\left\{\left\{\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&a[t]\,a''[t]&0&0\\-a[t]\,a''[t]&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&a[t]\,\mathrm{Sin}[\chi]^2\,a''[t]&0\\0&0&0&0\\-a[t]\,\mathrm{Sin}[\chi]^2\,a''[t]&0&0&0\\0&0&0&0\end{pmatrix},\right.\right.$$

$$\left.\begin{pmatrix}0&0&0&a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a''[t]\\0&0&0&0\\0&0&0&0\\-a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a''[t]&0&0&0\end{pmatrix}\right\},$$

$$\left\{\begin{pmatrix}0&\frac{a''[t]}{a[t]}&0&0\\-\frac{a''[t]}{a[t]}&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0\\0&-\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0&0\\0&0&0&0\end{pmatrix},\right.$$

$$\left.\begin{pmatrix}0&0&0&0\\0&0&0&\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)\\0&0&0&0\\0&-\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0&0\end{pmatrix}\right\},$$

$$\left\{\begin{pmatrix}0&0&\frac{a''[t]}{a[t]}&0\\0&0&0&0\\-\frac{a''[t]}{a[t]}&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&-1-a'[t]^2&0\\0&1+a'[t]^2&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix},\right.$$

$$\left.\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)\\0&0&-\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0\end{pmatrix}\right\}, \left\{\begin{pmatrix}0&0&0&\frac{a''[t]}{a[t]}\\0&0&0&0\\0&0&0&0\\-\frac{a''[t]}{a[t]}&0&0&0\end{pmatrix},\right.$$

$$\left.\left.\begin{pmatrix}0&0&0&0\\0&0&0&-1-a'[t]^2\\0&0&0&0\\0&1+a'[t]^2&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&-\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)\\0&0&\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}\right\}\right\}$$

Check that: $R^{\mu}{}_{\nu\rho\sigma} = -R^{\mu}{}_{\nu\sigma\rho}$

In[ ]:=
```
Table[riemann[[i, j]] // MatrixForm, {i, n}, {j, n}] // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix}
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0 & a[t]\,a''[t] & 0 & 0\\ -a[t]\,a''[t] & 0 & 0 & 0\\ 0&0&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0 & 0 & a[t]\,\mathrm{Sin}[\chi]^2\,a''[t] & 0\\ 0&0&0&0\\ -a[t]\,\mathrm{Sin}[\chi]^2\,a''[t] & 0&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} \\ \\ \\ - \end{pmatrix} \\
\begin{pmatrix} 0 & \frac{a''[t]}{a[t]} & 0 & 0\\ -\frac{a''[t]}{a[t]} & 0 & 0 & 0\\ 0&0&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\ 0&0&\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0\\ 0&-\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0\\0\\0\\0 \end{pmatrix} \\
\begin{pmatrix} 0&0&\frac{a''[t]}{a[t]}&0\\ 0&0&0&0\\ -\frac{a''[t]}{a[t]}&0&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\ 0&0&-1-a'[t]^2&0\\ 0&1+a'[t]^2&0&0\\ 0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0\\0\\0\\0 \end{pmatrix} \\
\begin{pmatrix} 0&0&0&\frac{a''[t]}{a[t]}\\ 0&0&0&0\\ 0&0&0&0\\ -\frac{a''[t]}{a[t]}&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\ 0&0&0&-1-a'[t]^2\\ 0&0&0&0\\ 0&1+a'[t]^2&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\ 0&0&0&0\\ 0&0&0&-\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)\\ 0&0&\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)&0 \end{pmatrix} &
\end{pmatrix}$$

riemann[[i,j]] gives the matrix $\left(R^i{}_j\right)_{kl}$

e.g. {{a,b,c},{d,e,f},{g,h,i}}[[2,3]] → f (2→row 3→column)

In[ ]:=
```
riemann[[1, 2, 2, 1]]
```

Out[ ]=

$-a[t]\,a''[t]$

Now we calculate the $R_{\mu\nu\rho\lambda} = g_{\mu\sigma}\,R^\sigma{}_{\nu\rho\lambda}$

In[ ]:=
```
lriemann := lriemann = FullSimplify[Table[
     Sum[metric[[i, ii]] riemann[[ii, j, k, l]], {ii, 1, n}],
     {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listlriemann := Table[
   If[
    UnsameQ[lriemann[[i, j, k, l]], 0],
    {Subscript[R, i, j, k, l], lriemann[[i, j, k, l]]}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
TableForm[Partition[DeleteCases[Flatten[listlriemann], Null], 2],
 TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

$R_{2,1,2,1}$	$-a[t]\, a''[t]$
$R_{3,1,3,1}$	$-a[t]\, \mathrm{Sin}[\chi]^2\, a''[t]$
$R_{3,2,3,2}$	$a[t]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$
$R_{4,1,4,1}$	$-a[t]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, a''[t]$
$R_{4,2,4,2}$	$a[t]^2\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2)$
$R_{4,3,4,3}$	$a[t]^2\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^4\, (1 + a'[t]^2)$

Check the following antisymmetry properties: $R_{\mu\nu\,\rho\sigma} = -R_{\mu\nu\,\sigma\rho} = -R_{\nu\mu\,\rho\sigma}$

In[ ]:=
```
Table[lriemann[[i, j]] // MatrixForm, {i, n}, {j, n}] // MatrixForm
```

Out[ ]//MatrixForm=

$$\left(\begin{array}{ll}
\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & -a[t]\, a'' \\ a[t]\, a''[t] & 0 \\ 0 & 0 \\ 0 & 0 \end{pmatrix} \\
\begin{pmatrix} 0 & a[t]\, a''[t] & 0 & 0 \\ -a[t]\, a''[t] & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \\
\begin{pmatrix} 0 & 0 & a[t]\, \mathrm{Sin}[\chi]^2\, a''[t] & 0 \\ 0 & 0 & 0 & 0 \\ -a[t]\, \mathrm{Sin}[\chi]^2\, a''[t] & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & 0 & \\ 0 & 0 & -a[t] \\ 0 & a[t]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2) & \\ 0 & 0 & \end{pmatrix} \\
\begin{pmatrix} 0 & 0 & 0 & a[t]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, a''[t] \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -a[t]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, a''[t] & 0 & 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & 0 & 0 & \\ 0 & 0 & 0 & -\varepsilon \\ 0 & 0 & 0 & \\ 0 & a[t]^2\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, (1 + a'[t]^2) & 0 & \end{pmatrix}
\end{array}\right.$$

Now calculate contravariant Riemann: $R^{\mu\nu\rho\lambda} = g^{\nu\alpha}\, g^{\rho\beta}\, g^{\lambda\gamma}\, R^{\mu}{}_{\alpha\beta\gamma}$

In[ ]:=
```
uriemann := uriemann = FullSimplify[Table[
      Sum[
       inversemetric[[j, jj]] inversemetric[[k, kk]] inversemetric[[l, ll]]
        riemann[[i, jj, kk, ll]], {jj, 1, n}, {kk, 1, n}, {ll, 1, n}
      ],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listuriemann := Table[
   If[
    UnsameQ[uriemann[[i, j, k, l]], 0], {Superscript[
      Superscript[Superscript[Superscript[R, i], j], k], l], uriemann[[i, j, k, l]]}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
TableForm[Partition[DeleteCases[Flatten[listuriemann], Null], 2],
 TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

$R^{2^{1^{2^{1}}}}$	$-\frac{a''[t]}{a[t]^3}$
$R^{3^{1^{3^{1}}}}$	$-\frac{Csc[\chi]^2\, a''[t]}{a[t]^3}$
$R^{3^{2^{3^{2}}}}$	$\frac{Csc[\chi]^2\,(1+a'[t]^2)}{a[t]^6}$
$R^{4^{1^{4^{1}}}}$	$-\frac{Csc[\theta]^2\, Csc[\chi]^2\, a''[t]}{a[t]^3}$
$R^{4^{2^{4^{2}}}}$	$\frac{Csc[\theta]^2\, Csc[\chi]^2\,(1+a'[t]^2)}{a[t]^6}$
$R^{4^{3^{4^{3}}}}$	$\frac{Csc[\theta]^2\, Csc[\chi]^4\,(1+a'[t]^2)}{a[t]^6}$

Calculating and displaying the R^2 scalar:

Now calculate $R^2 = R_{\mu\nu\lambda\rho}\, R^{\mu\nu\lambda\rho}$

In[ ]:=
```
r2 = FullSimplify[
  Sum[lriemann[[i, j, k, l]] uriemann[[i, j, k, l]], {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]]
```

Out[ ]=

$$\frac{12\left(\left(1 + a'[t]^2\right)^2 + a[t]^2\, a''[t]^2\right)}{a[t]^4}$$

Calculating and displaying the Ricci tensor:

The Ricci tensor $R_{\mu\nu}$ was defined by summing the first and third indices of the Riemann tensor (which has the first index already raised).

```
ricci := ricci = FullSimplify[Table[
     Sum[
      riemann[[i, j, i, l]],
      {i, 1, n}
     ], {j, 1, n}, {l, 1, n}]];
listricci := Table[
   If[
    UnsameQ[ricci[[j, l]], 0],
    {Subscript[R, j, l], ricci[[j, l]]}
   ] , {j, 1, n}, {l, 1, j}];
TableForm[Partition[DeleteCases[Flatten[listricci], Null], 2], TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

$R_{1,1}$	$-\frac{3\,a''[t]}{a[t]}$
$R_{2,2}$	$2+2\,a'[t]^2+a[t]\,a''[t]$
$R_{3,3}$	$\mathrm{Sin}[\chi]^2\,(2+2\,a'[t]^2+a[t]\,a''[t])$
$R_{4,4}$	$\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,(2+2\,a'[t]^2+a[t]\,a''[t])$

In[ ]:=

```
ricci // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{3\,a''[t]}{a[t]} & 0 & 0 & 0 \\ 0 & 2+2\,a'[t]^2+a[t]\,a''[t] & 0 & 0 \\ 0 & 0 & \mathrm{Sin}[\chi]^2\,(2+2\,a'[t]^2+a[t]\,a''[t]) & 0 \\ 0 & 0 & 0 & \mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,(2+2\,a'[t]^2+a[t]\,a''[t]) \end{pmatrix}$$

A vanishing table (as with the Schwarzschild metric example) means that the vacuum Einstein equation is satisfied.

Calculating the scalar curvature:

The scalar curvature *R* is calculated using the inverse metric and the Ricci tensor. The result is displayed in the output line.

In[ ]:=

```
scalar = FullSimplify[Sum[inversemetric[[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}]]
```

Out[ ]=

$$\frac{6\,(1+a'[t]^2+a[t]\,a''[t])}{a[t]^2}$$

Calculating the Einstein tensor:

The Einstein tensor, $G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2}\,g_{\mu\nu}\,R$, is found from the tensors already calculated.

In[ ]:=
```
einstein := einstein = FullSimplify[ricci - (1/2) scalar * metric];
listeinstein := Table[
   If[
    UnsameQ[einstein[[j, l]], 0],
    {Subscript[G, j, l], einstein[[j, l]]}
   ], {j, 1, n}, {l, 1, j}];
TableForm[Partition[DeleteCases[Flatten[listeinstein], Null], 2],
 TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

$G_{1,1}$	$\frac{3\,(1+a'[t]^2)}{a[t]^2}$
$G_{2,2}$	$-1-a'[t]^2-2\,a[t]\,a''[t]$
$G_{3,3}$	$-\mathrm{Sin}[\chi]^2\,(1+a'[t]^2+2\,a[t]\,a''[t])$
$G_{4,4}$	$-\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,(1+a'[t]^2+2\,a[t]\,a''[t])$

In[ ]:=
```
einstein // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{3\,(1+a'[t]^2)}{a[t]^2} & 0 & 0 & 0 \\ 0 & -1-a'[t]^2-2\,a[t]\,a''[t] & 0 & 0 \\ 0 & 0 & -\mathrm{Sin}[\chi]^2\,(1+a'[t]^2+2\,a[t]\,a''[t]) & 0 \\ 0 & 0 & 0 & -\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,(1+a'[t]^2+2\,a[t]\,a \end{pmatrix}$$

A vanishing table means that the vacuum Einstein equation is satisfied!

Calculating the Weyl tensor:

The Weyl tensor:

$$C_{\rho\sigma\mu\nu} = R_{\rho\sigma\mu\nu} - \frac{2}{n-2}\left\{g_{\rho[\mu}R_{\nu]\sigma} - g_{\sigma[\mu}R_{\nu]\rho}\right\} + \frac{2}{(n-1)(n-2)}\,g_{\rho[\mu}\,g_{\nu]\sigma}\quad R$$

$$C_{ijkl} = R_{ijkl} - \frac{1}{n-2}\left\{g_{ik}R_{lj} - g_{il}R_{kj} - g_{jk}R_{li} + g_{jl}R_{ki}\right\} + \frac{1}{(n-1)(n-2)}\left\{g_{ik}\,g_{lj} - g_{il}\,g_{kj}\right\}R$$

In[ ]:=

```
weyl := weyl = FullSimplify[Table[
      If[n > 3,
       lriemann[[i, j, k, l]]
        -    1/(n - 2)    (metric[[i, k]] ricci[[l, j]] - metric[[i, l]] ricci[[k, j]] -
            metric[[j, k]] ricci[[l, i]] + metric[[j, l]] ricci[[k, i]])
        + 1/((n - 1) (n - 2)) (metric[[i, k]] metric[[l, j]] - metric[[i, l]] metric[[k, j]]) scalar
       (*else, if n≤ 3 return 0:*), 0],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listweyl := Table[
    If[
     UnsameQ[weyl[[i, j, k, l]], 0],
     {Subscript[C, i, j, k, l], weyl[[i, j, k, l]]}
    ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
TableForm[Partition[DeleteCases[Flatten[listweyl], Null], 2], TableSpacing → {2, 2}]
```

Out[ ]//TableForm=

```
{}
```

Calculating the geodesic equations:

The geodesic equations are calculated by asking *Mathematica* to carry out the sum $-\Gamma^{\alpha}{}_{\beta\gamma} u^{\beta} u^{\gamma}$, where u^{α} are the components of the four-velocity. (This gives the derivative of u^{α} with respect to proper time τ. (This is replaced by s if the geodesics are spacelike.)

In[ ]:=

```
geodesic := geodesic = Simplify[Table[
      -Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n},
        {k, 1, n}], {i, 1, n}]];
listgeodesic := Table[{"d/dτ" ToString[u[i]], "=", geodesic[[i]]}, {i, 1, n}];
TableForm[listgeodesic, TableSpacing → {2}]
```

Out[ ]//TableForm=

d/dτ u[1]	=	$-a[t]\,(u[2]^2 + \mathrm{Sin}[\chi]^2\,(u[3]^2 + \mathrm{Sin}[\theta]^2\,u[4]^2))\,a'[t]$
d/dτ u[2]	=	$\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi]\,(u[3]^2 + \mathrm{Sin}[\theta]^2\,u[4]^2) - \frac{2\,u[1]\,u[2]\,a'[t]}{a[t]}$
d/dτ u[3]	=	$-2\,\mathrm{Cot}[\chi]\,u[2]\,u[3] + \mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,u[4]^2 - \frac{2\,u[1]\,u[3]\,a'[t]}{a[t]}$
d/dτ u[4]	=	$-\frac{2\,u[4]\,(a[t]\,(\mathrm{Cot}[\chi]\,u[2]+\mathrm{Cot}[\theta]\,u[3])+u[1]\,a'[t])}{a[t]}$

And here they are displayed as differential equations for the coordinates. τ is the affine parameter.

In[ ]:=
```
geodesic := geodesic =
   Simplify[Table[-Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n}, {k, 1, n}], {i, 1, n}]];
subst = Table[u[i] → Subscript[coord[[i]], τ], {i, 1, n}];
nlistgeodesic :=
  Table[{Subscript[coord[[i]], ττ], "+", -geodesic[[i]] /. subst, "= 0"}, {i, 1, n}];
TableForm[nlistgeodesic, TableSpacing → {2}]
```

Out[ ]//TableForm=

$t_{\tau\tau}$	+	$a[t]\left(\mathrm{Sin}[\chi]^2\left(\theta_\tau^2+\mathrm{Sin}[\theta]^2\,\phi_\tau^2\right)+\chi_\tau^2\right)a'[t]$	= 0
$\chi_{\tau\tau}$	+	$-\mathrm{Cos}[\chi]\,\mathrm{Sin}[\chi]\left(\theta_\tau^2+\mathrm{Sin}[\theta]^2\,\phi_\tau^2\right)+\frac{2\,t_\tau\,\chi_\tau\,a'[t]}{a[t]}$	= 0
$\theta_{\tau\tau}$	+	$-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\phi_\tau^2+2\,\mathrm{Cot}[\chi]\,\theta_\tau\,\chi_\tau+\frac{2\,t_\tau\,\theta_\tau\,a'[t]}{a[t]}$	= 0
$\phi_{\tau\tau}$	+	$\frac{2\,\phi_\tau\,(a[t]\,(\mathrm{Cot}[\theta]\,\theta_\tau+\mathrm{Cot}[\chi]\,\chi_\tau)+t_\tau\,a'[t])}{a[t]}$	= 0

The Sphere S^2

We use coordinates (θ, ϕ), and the metric:

$$ds^2 = d\theta^2 + \sin^2(\theta)\, d\phi^2$$

```
Clear[coord, metric, inversemetric, affine, riemann, lriemann, uriemann,
  ricci, scalar, einstein, weyl, geodesic, R, G, τ, i, j, k, l, s, m];
Clear[r, θ, ϕ, t, χ, a];

(*-------------------------------------------------------------------------*)
(* This is what you need to set: *)
coord = {θ, ϕ};
n     = Length[coord];
metric = {
    {1, 0      },
    {0, Sin[θ]^2}
  };
(*-------------------------------------------------------------------------*)

inversemetric = FullSimplify[Inverse[metric]];
Print["--------------------------------------------------------------------"];
Print["The Manifold has dimension   n= ",
  n, "\nCoordinate system:           ", coord];
Print["--------------------------------------------------------------------"];
Print["gμν=", metric // MatrixForm];
Print["g^μν=", inversemetric // MatrixForm];
affine := affine = FullSimplify[Table[
```

```
      (1/2)*Sum[
        (*                gis  (∂kgsj+∂jgsk−∂sgjk)            *)
        (inversemetric[[i, s]])*
         (D[metric[[s, j]], coord[[k]]] +
            D[metric[[s, k]], coord[[j]]] - D[metric[[j, k]], coord[[s]]]),
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}]];
(*The non zero Christoffel symbols are computed and selected below: *)
listaffine := Table[
   If[
    UnsameQ[affine[[i, j, k]], 0],
    {Subscript[Superscript[Γ, i], j, k], affine[[i, j, k]]}
     ],
   {i, 1, n}, {j, 1, n}, {k, 1, j}];
Print["---------------------------------------------------------------------"];
Print["Christoffel Symbols:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listaffine], Null], 2], TableSpacing → {2, 2}]];
riemann := riemann = FullSimplify[Table[
      (* Rijkl=     ∂kΓilj                          -           ∂lΓikj*)
      D[   affine[[i, l, j]],  coord[[k]]     ] - D[affine[[i, k, j]], coord[[l]]    ] +
       (*          Γiks                 Γslj      -          Γils                Γskj      *)
       Sum[affine[[i, k, s]] affine[[s, l, j]] -  affine[[i, l, s]] affine[[s, k, j]] ,
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listriemann := Table[
   If[
    UnsameQ[riemann[[i, j, k, l]], 0],
    {Subscript[Superscript[R, i], j, k, l], riemann[[i, j, k, l]]}
   ],
   {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}];
Print["---------------------------------------------------------------------"];
Print["Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listriemann], Null], 2], TableSpacing → {2, 2}]];
lriemann := lriemann = FullSimplify[Table[
      Sum[metric[[i, ii]] riemann[[ii, j, k, l]], {ii, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listlriemann := Table[
   If[
    UnsameQ[lriemann[[i, j, k, l]], 0],
    {Subscript[R, i, j, k, l], lriemann[[i, j, k, l]]}
```

```
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------"];
Print["Contravariant Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listlriemann], Null], 2], TableSpacing → {2, 2}]];
uriemann := uriemann = FullSimplify[Table[
     Sum[
      inversemetric⟦j, jj⟧ inversemetric⟦k, kk⟧ inversemetric⟦l, ll⟧
       riemann⟦i, jj, kk, ll⟧, {jj, 1, n}, {kk, 1, n}, {ll, 1, n}
     ],
     {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listuriemann := Table[
   If[
    UnsameQ[uriemann⟦i, j, k, l⟧, 0], {Superscript[
      Superscript[Superscript[Superscript[R, i], j], k], l], uriemann⟦i, j, k, l⟧}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------"];
Print["Covariant Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listuriemann], Null], 2], TableSpacing → {2, 2}]];
r2 = FullSimplify[Sum[lriemann⟦i, j, k, l⟧ uriemann⟦i, j, k, l⟧,
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
Print["-----------------------------------------------------------------"];
Print["R^2= ", r2];
ricci := ricci = FullSimplify[Table[
     Sum[
      riemann⟦i, j, i, l⟧,
      {i, 1, n}
     ], {j, 1, n}, {l, 1, n}]];
listricci := Table[
   If[
    UnsameQ[ricci⟦j, l⟧, 0],
    {Subscript[R, j, l], ricci⟦j, l⟧}
   ], {j, 1, n}, {l, 1, j}];
Print["-----------------------------------------------------------------"];
Print["Ricci Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listricci], Null], 2], TableSpacing → {2, 2}]];
scalar = FullSimplify[Sum[inversemetric⟦i, j⟧ ricci⟦i, j⟧, {i, 1, n}, {j, 1, n}]];
Print["-----------------------------------------------------------------"];
Print["Curvature Scalar:"];
Print["R= ", scalar];
einstein := einstein = FullSimplify[ricci - (1/2) scalar * metric];
```

```
listeinstein := Table[
   If[
    UnsameQ[einstein[[j, l]], 0],
    {Subscript[G, j, l], einstein[[j, l]]}
   ], {j, 1, n}, {l, 1, j}];
Print["-----------------------------------------------------------------------"];
Print["Einstein Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listeinstein], Null], 2], TableSpacing → {2, 2}]];
weyl := weyl = FullSimplify[Table[
      If[n > 3,
       lriemann[[i, j, k, l]]
        -     1/(n - 2)     (metric[[i, k]] ricci[[l, j]] - metric[[i, l]] ricci[[k, j]] -
            metric[[j, k]] ricci[[l, i]] + metric[[j, l]] ricci[[k, i]])
        + 1/((n - 1) (n - 2)) (metric[[i, k]] metric[[l, j]] - metric[[i, l]] metric[[k, j]]) scalar
       (*else, if n≤ 3 return 0:*), 0],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listweyl := Table[
   If[
    UnsameQ[weyl[[i, j, k, l]], 0],
    {Subscript[C, i, j, k, l], weyl[[i, j, k, l]]}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Weyl Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listweyl], Null], 2], TableSpacing → {2, 2}]];
geodesic := geodesic =
   Simplify[Table[-Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n}, {k, 1, n}], {i, 1, n}]];
subst = Table[u[i] → Subscript[coord[[i]], τ], {i, 1, n}];
nlistgeodesic :=
  Table[{Subscript[coord[[i]], ττ], "+", -geodesic[[i]] /. subst, "= 0"}, {i, 1, n}];
Print["-----------------------------------------------------------------------"];
Print["Geodesic Equations:"];
Print[TableForm[nlistgeodesic, TableSpacing → {2}]];
```

```
-----------------------------------------------------------------------

The Manifold has dimension   n= 2
Coordinate system:           {θ, ϕ}

-----------------------------------------------------------------------
```

$g_{\mu\nu}=\begin{pmatrix} 1 & 0 \\ 0 & \text{Sin}[\theta]^2 \end{pmatrix}$

$g^{\mu\nu}=\begin{pmatrix} 1 & 0 \\ 0 & \text{Csc}[\theta]^2 \end{pmatrix}$

Christoffel Symbols:

$\Gamma^{1}{}_{2,2}$ $-\text{Cos}[\theta]\,\text{Sin}[\theta]$

$\Gamma^{2}{}_{2,1}$ $\text{Cot}[\theta]$

Riemann Tensor:

$R^{1}{}_{2,2,1}$ $-\text{Sin}[\theta]^2$

$R^{2}{}_{1,2,1}$ 1

Contravariant Riemann Tensor:

$R_{2,1,2,1}$ $\text{Sin}[\theta]^2$

Covariant Riemann Tensor:

$R^{2^{1^{2^{1}}}}$ $\text{Csc}[\theta]^2$

$R^2= 4$

Ricci Tensor:

$R_{1,1}$ 1

$R_{2,2}$ $\text{Sin}[\theta]^2$

Curvature Scalar:

R= 2

Einstein Tensor:

{}

Weyl Tensor:

{}

Geodesic Equations:

$\theta_{\tau\tau}$ + $-\text{Cos}[\theta]\,\text{Sin}[\theta]\,\phi_\tau^2$ = 0

$\phi_{\tau\tau}$ + $2\,\text{Cot}[\theta]\,\theta_\tau\,\phi_\tau$ = 0

The Schwarzschild Metric

We consider the Schwarzschild metric outside the horizon. We use (t, r, θ, ϕ) coordinates, and the line element:

$$ds^2 = -\left(1 - \frac{2m}{r}\right) dt^2 + \left(1 - \frac{2m}{r}\right)^{-1} dr^2 + r^2\left(d\theta^2 + \sin^2(\theta)\, d\phi^2\right)$$

```
Clear[coord, metric, inversemetric, affine, riemann, lriemann, uriemann,
  ricci, scalar, einstein, weyl, geodesic, R, G, τ, i, j, k, l, s, m];
Clear[r, θ, ϕ, t, χ, a];

(*---------------------------------------------------------------------------*)
(* This is what you need to set: *)
coord = {t, r, θ, ϕ};
n     = Length[coord];
metric = {
   {-(1 - 2 m/r), 0          , 0   ,            0},
   {0           , 1/(1 - 2 m/r), 0   ,            0},
   {0           , 0          , r^2 ,            0},
   {0           , 0          , 0   , r^2 Sin[θ]^2}};
(*---------------------------------------------------------------------------*)

inversemetric = FullSimplify[Inverse[metric]];
Print["---------------------------------------------------------------------------"];
Print["The Manifold has dimension    n= ",
  n, "\nCoordinate system:            ", coord];
Print["---------------------------------------------------------------------------"];
Print["gμν=", metric // MatrixForm];
Print["gμν=", inversemetric // MatrixForm];
affine := affine = FullSimplify[Table[
      (1 / 2) * Sum[
        (*              g^is (∂k gsj+∂j gsk-∂s gjk)              *)
        (inversemetric[[i, s]]) *
         (D[metric[[s, j]], coord[[k]]] +
           D[metric[[s, k]], coord[[j]]] - D[metric[[j, k]], coord[[s]]]),
```

```
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}]];
(*The non zero Christoffel symbols are computed and selected below: *)
listaffine := Table[
   If[
    UnsameQ[affine[[i, j, k]], 0],
    {Subscript[Superscript[Γ, i], j, k], affine[[i, j, k]]}
     ],
   {i, 1, n}, {j, 1, n}, {k, 1, j}];
Print["-----------------------------------------------------------------------"];
Print["Christoffel Symbols:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listaffine], Null], 2], TableSpacing → {2, 2}]];
riemann := riemann = FullSimplify[Table[
      (* R^i_jkl=     ∂_kΓ^i_lj                          -          ∂_lΓ^i_kj*)
      D[   affine[[i, l, j]],  coord[[k]]     ] - D[affine[[i, k, j]], coord[[l]]    ] +
       (*          Γ^i_ks                  Γ^s_lj      -          Γ^i_ls                Γ^s_kj       *)
       Sum[affine[[i, k, s]] affine[[s, l, j]] -  affine[[i, l, s]] affine[[s, k, j]] ,
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listriemann := Table[
   If[
    UnsameQ[riemann[[i, j, k, l]], 0],
    {Subscript[Superscript[R, i], j, k, l], riemann[[i, j, k, l]]}
   ],
   {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listriemann], Null], 2], TableSpacing → {2, 2}]];
lriemann := lriemann = FullSimplify[Table[
      Sum[metric[[i, ii]] riemann[[ii, j, k, l]], {ii, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listlriemann := Table[
   If[
    UnsameQ[lriemann[[i, j, k, l]], 0],
    {Subscript[R, i, j, k, l], lriemann[[i, j, k, l]]}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Contravariant Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listlriemann], Null], 2], TableSpacing → {2, 2}]];
```

```
uriemann := uriemann = FullSimplify[Table[
     Sum[
      inversemetric[[j, jj]] inversemetric[[k, kk]] inversemetric[[l, ll]]
       riemann[[i, jj, kk, ll]], {jj, 1, n}, {kk, 1, n}, {ll, 1, n}
     ],
     {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listuriemann := Table[
   If[
    UnsameQ[uriemann[[i, j, k, l]], 0], {Superscript[
      Superscript[Superscript[Superscript[R, i], j], k], l], uriemann[[i, j, k, l]]}
   ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["----------------------------------------------------------------------------"];
Print["Covariant Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listuriemann], Null], 2], TableSpacing → {2, 2}]];
r2 = FullSimplify[Sum[lriemann[[i, j, k, l]] uriemann[[i, j, k, l]],
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
Print["----------------------------------------------------------------------------"];
Print["R²= ", r2];
ricci := ricci = FullSimplify[Table[
     Sum[
      riemann[[i, j, i, l]],
      {i, 1, n}
     ], {j, 1, n}, {l, 1, n}]];
listricci := Table[
   If[
    UnsameQ[ricci[[j, l]], 0],
    {Subscript[R, j, l], ricci[[j, l]]}
   ], {j, 1, n}, {l, 1, j}];
Print["----------------------------------------------------------------------------"];
Print["Ricci Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listricci], Null], 2], TableSpacing → {2, 2}]];
scalar = FullSimplify[Sum[inversemetric[[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}]];
Print["----------------------------------------------------------------------------"];
Print["Curvature Scalar:"];
Print["R= ", scalar];
einstein := einstein = FullSimplify[ricci - (1 / 2) scalar * metric];
listeinstein := Table[
   If[
    UnsameQ[einstein[[j, l]], 0],
    {Subscript[G, j, l], einstein[[j, l]]}
   ], {j, 1, n}, {l, 1, j}];
```

```
Print["--------------------------------------------------------------------"];
Print["Einstein Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listeinstein], Null], 2], TableSpacing → {2, 2}]];
weyl := weyl = FullSimplify[Table[
      If[n > 3,
       lriemann[[i, j, k, l]]
        -      1/(n - 2)    (metric[[i, k]] ricci[[l, j]] - metric[[i, l]] ricci[[k, j]] -
            metric[[j, k]] ricci[[l, i]] + metric[[j, l]] ricci[[k, i]])
        +  1/((n - 1) (n - 2)) (metric[[i, k]] metric[[l, j]] - metric[[i, l]] metric[[k, j]]) scalar
       (*else, if n≤ 3 return 0:*), 0],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listweyl := Table[
   If[
    UnsameQ[weyl[[i, j, k, l]], 0],
    {Subscript[C, i, j, k, l], weyl[[i, j, k, l]]}
   ] , {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["--------------------------------------------------------------------"];
Print["Weyl Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listweyl], Null], 2], TableSpacing → {2, 2}]];
geodesic := geodesic =
   Simplify[Table[-Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n}, {k, 1, n}], {i, 1, n}]];
subst = Table[u[i] → Subscript[coord[[i]], τ], {i, 1, n}];
nlistgeodesic :=
  Table[{Subscript[coord[[i]], ττ], "+", -geodesic[[i]] /. subst, "= 0"}, {i, 1, n}];
Print["--------------------------------------------------------------------"];
Print["Geodesic Equations:"];
Print[TableForm[nlistgeodesic, TableSpacing → {2}]];
```

```
--------------------------------------------------------------------

The Manifold has dimension   n= 4
Coordinate system:           {t, r, θ, ϕ}

--------------------------------------------------------------------
```

$$g_{\mu\nu}=\begin{pmatrix} -1+\frac{2m}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2m}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

$$g^{\mu\nu} = \begin{pmatrix} \frac{r}{2m-r} & 0 & 0 & 0 \\ 0 & 1-\frac{2m}{r} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2}{r^2} \end{pmatrix}$$

Christoffel Symbols:

$\Gamma^1{}_{2,1}$ $\frac{m}{r(-2m+r)}$

$\Gamma^2{}_{1,1}$ $\frac{m(-2m+r)}{r^3}$

$\Gamma^2{}_{2,2}$ $\frac{m}{2mr-r^2}$

$\Gamma^2{}_{3,3}$ $2m-r$

$\Gamma^2{}_{4,4}$ $(2m-r)\,\mathrm{Sin}[\theta]^2$

$\Gamma^3{}_{3,2}$ $\frac{1}{r}$

$\Gamma^3{}_{4,4}$ $-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]$

$\Gamma^4{}_{4,2}$ $\frac{1}{r}$

$\Gamma^4{}_{4,3}$ $\mathrm{Cot}[\theta]$

Riemann Tensor:

$R^1{}_{2,2,1}$ $-\frac{2m}{r^2(-2m+r)}$

$R^1{}_{3,3,1}$ $\frac{m}{r}$

$R^1{}_{4,4,1}$ $\frac{m\,\mathrm{Sin}[\theta]^2}{r}$

$R^2{}_{1,2,1}$ $\frac{2m(2m-r)}{r^4}$

$R^2{}_{3,3,2}$ $\frac{m}{r}$

$R^2{}_{4,4,2}$ $\frac{m\,\mathrm{Sin}[\theta]^2}{r}$

$R^3{}_{1,3,1}$ $\frac{m(-2m+r)}{r^4}$

$R^3{}_{2,3,2}$ $\frac{m}{(2m-r)r^2}$

$R^3{}_{4,4,3}$ $-\frac{2m\,\mathrm{Sin}[\theta]^2}{r}$

$R^4{}_{1,4,1}$ $\frac{m(-2m+r)}{r^4}$

$R^4{}_{2,4,2}$ $\frac{m}{(2m-r)r^2}$

$R^4{}_{3,4,3}$ $\frac{2m}{r}$

Contravariant Riemann Tensor:

$R_{2,1,2,1}$ $-\frac{2\,m}{r^3}$

$R_{3,1,3,1}$ $\frac{m\,(-2\,m+r)}{r^2}$

$R_{3,2,3,2}$ $\frac{m}{2\,m-r}$

$R_{4,1,4,1}$ $\frac{m\,(-2\,m+r)\,\mathrm{Sin}[\theta]^2}{r^2}$

$R_{4,2,4,2}$ $\frac{m\,\mathrm{Sin}[\theta]^2}{2\,m-r}$

$R_{4,3,4,3}$ $2\,m\,r\,\mathrm{Sin}[\theta]^2$

Covariant Riemann Tensor:

R^{2121} $-\frac{2\,m}{r^3}$

R^{3131} $\frac{m}{r^4\,(-2\,m+r)}$

R^{3232} $\frac{m\,(2\,m-r)}{r^6}$

R^{4141} $-\frac{m\,\mathrm{Csc}[\theta]^2}{(2\,m-r)\,r^4}$

R^{4242} $\frac{m\,(2\,m-r)\,\mathrm{Csc}[\theta]^2}{r^6}$

R^{4343} $\frac{2\,m\,\mathrm{Csc}[\theta]^2}{r^7}$

$R^2 = \frac{48\,m^2}{r^6}$

Ricci Tensor:

{}

Curvature Scalar:

R= 0

Einstein Tensor:

{}

Weyl Tensor:

$C_{2,1,2,1}$	$-\frac{2\,m}{r^3}$
$C_{3,1,3,1}$	$\frac{m\,(-2\,m+r)}{r^2}$
$C_{3,2,3,2}$	$\frac{m}{2\,m-r}$
$C_{4,1,4,1}$	$\frac{m\,(-2\,m+r)\,\text{Sin}[\theta]^2}{r^2}$
$C_{4,2,4,2}$	$\frac{m\,\text{Sin}[\theta]^2}{2\,m-r}$
$C_{4,3,4,3}$	$2\,m\,r\,\text{Sin}[\theta]^2$

Geodesic Equations:

$$t_{\tau\tau} + -\frac{2\,m\,r_\tau\,t_\tau}{2\,m\,r-r^2} = 0$$

$$r_{\tau\tau} + \frac{m\,r_\tau^2}{2\,m\,r-r^2} - \frac{m\,(2\,m-r)\,t_\tau^2}{r^3} + (2\,m-r)\,\theta_\tau^2 + (2\,m-r)\,\text{Sin}[\theta]^2\,\phi_\tau^2 = 0$$

$$\theta_{\tau\tau} + \frac{2\,r_\tau\,\theta_\tau}{r} - \text{Cos}[\theta]\,\text{Sin}[\theta]\,\phi_\tau^2 = 0$$

$$\phi_{\tau\tau} + \frac{2\,(r_\tau + r\,\text{Cot}[\theta]\,\theta_\tau)\,\phi_\tau}{r} = 0$$

We can see the following symmetries of the Weyl tensor in this display: $C_{\mu\nu\,\rho\sigma} = -C_{\mu\nu\,\sigma\rho} = -C_{\nu\mu\,\rho\sigma}$

In[ ]:=
```
Table[weyl[[i, j]] // MatrixForm, {i, n}, {j, n}] // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix}
\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&-\frac{2m}{r^3}&0&0\\\frac{2m}{r^3}&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&\frac{m(-2m+r)}{r^2}&0\\0&0&0&0\\\frac{m(2m-r)}{r^2}&0&0&0\\0&0&0&0\end{pmatrix} \\
\begin{pmatrix}0&\frac{2m}{r^3}&0&0\\-\frac{2m}{r^3}&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&\frac{m}{2m-r}&0\\0&\frac{m}{-2m+r}&0&0\\0&0&0&0\end{pmatrix} \\
\begin{pmatrix}0&0&\frac{m(2m-r)}{r^2}&0\\0&0&0&0\\\frac{m(-2m+r)}{r^2}&0&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&\frac{m}{-2m+r}&0\\0&\frac{m}{2m-r}&0&0\\0&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix} \\
\begin{pmatrix}0&0&0&\frac{m(2m-r)\text{Sin}[\theta]^2}{r^2}\\0&0&0&0\\0&0&0&0\\\frac{m(-2m+r)\text{Sin}[\theta]^2}{r^2}&0&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&0&\frac{m\,\text{Sin}[\theta]^2}{-2m+r}\\0&0&0&0\\0&\frac{m\,\text{Sin}[\theta]^2}{2m-r}&0&0\end{pmatrix} &
\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&-2\,m\,r\,\text{Sin}[\theta]^2\\0&0&2\,m\,r\,\text{Sin}[\theta]^2&0\end{pmatrix}
\end{pmatrix}$$

Acknowledgment

The original program was kindly written by *Leonard Parker, University of Wisconsin, Milwaukee* especially for this text.

Several additions by Konstantinos Anagnostopoulos, National Tech U. Athens, http://physics.ntua.gr/konstant, konstant@mail.ntua.gr, 2023